

SPERRY RAND

# UNIVAC

FUNDAMENTALS  
OF COBOL

**REPORT WRITER**

PROGRAMMER  
REFERENCE

This document contains the latest information available at the time of publication. However, the Univac Division reserves the right to modify or revise its contents. To ensure that you have the most recent information, contact your local Univac Representative.

UNIVAC is a registered trademark of the Sperry Rand Corporation.

## CONTENTS

CONTENTS	1 to 2
PREFACE	1 to 1
1. THE REPORT	1-1 to 1-4
1.1. GENERAL	1-1
1.2. PARTS OF A REPORT	1-2
1.2.1. Page Headings, Footings, and Detail Data	1-2
1.2.2. Control Headings and Control Footings	1-3
1.2.3. Report Heading and Report Footing	1-4
2. REPORT WRITER IN A COBOL PROGRAM	2-1 to 2-4
2.1. GENERAL	2-1
2.2. REPORT WRITER ENTRIES	2-1
3. USING THE REPORT WRITER	3-1 to 3-16
3.1. GENERAL	3-1
3.2. DATA DIVISION ENTRIES	3-1
3.2.1. Report Description (RD)	3-2
3.2.2. Sample Report Group Description Entry	3-3
3.3. THE HEADING AND FOOTING	3-4
3.4. THE BODY OF A REPORT PAGE	3-7
3.4.1. The Detail Report Group	3-8
3.4.1.1. Multiple Detail Report Groups	3-8
3.4.1.2. Group Indicate	3-9
3.5. CONTROL HEADING AND FOOTING REPORT GROUPS	3-9
3.5.1. Summing Methods	3-11
3.5.2. The Sum Counter Reset Option	3-15
3.6. FIXED DATA NAMES	3-15
3.7. PROCEDURE DIVISION	3-15
4. SAMPLE PROGRAM	4-1 to 4-5
4.1. GENERAL	4-1
4.2. THE PROGRAM	4-1

<b>APPENDIX A. REPORT WRITER FORMATS</b>	A-1 to A-16
A.1. GENERAL	A-1
A.2. DATA DIVISION	A-1
A.2.1. File Description	A-1
A.2.1.1. Standard COBOL Clauses	A-2
A.2.1.2. REPORT Clause	A-2
A.2.1.3. The Report Section	A-2
A.2.1.4. Report Description Entries	A-3
A.2.1.4.1. CODE Clause	A-3
A.2.1.4.2. CONTROL Clause	A-4
A.2.1.4.3. PAGE LIMIT Clause	A-4
A.2.2. Report Group Entries	A-6
A.2.2.1. Data-name Clause	A-7
A.2.2.2. LINE NUMBER Clause	A-7
A.2.2.3. NEXT GROUP Clause	A-8
A.2.2.4. TYPE IS Clause	A-9
A.2.2.5. COLUMN NUMBER Clause	A-11
A.2.2.6. GROUP INDICATE Clause	A-11
A.2.2.7. RESET Clause	A-11
A.2.2.8. SOURCE, SUM, and VALUE Clauses	A-12
A.2.2.9. LINE-COUNTER and PAGE-COUNTER Fixed Data-names	A-13
A.2.3. Procedure Division	A-14
A.2.3.1. GENERATE Statement	A-14
A.2.3.2. INITIATE Statement	A-15
A.2.3.3. TERMINATE Statement	A-15
A.2.3.4. USE BEFORE REPORTING Clause	A-16
<b>FIGURES</b>	
1-1. Basic Elements of a Report	1-2
1-2. Control Headings and Control Footings	1-3
1-3. The Complete Report	1-4
2-1. Report Writer Source Coding	2-3
3-1. Sample RD	3-2
3-2. Report Group and Printed Line Relationship	3-5
3-3. Using the GROUP INDICATE Clause	3-10
3-4. Running and Crossfoot Sums	3-12
3-5. Coding for Sample Report	3-13
3-6. Sample Report	3-14
3-7. Use of Report Writer Verbs	3-16
4-1. Report Writer Program	4-3
4-2. The Report	4-4
<b>TABLES</b>	
3-1. Report Group Entries	3-4

## PREFACE

This manual is another in the series of manuals entitled "Fundamentals of COBOL." As with the other volumes of the series, it does not represent the COBOL implementation for any particular computer system; rather, it is intended as a basic reference source to acquaint the reader with the COBOL Report Writer feature. The information in this manual is in accordance with the Report Writer module, Level 2 of American National Standard COBOL, X3.23-1968.

The Report Writer module provides the facility for producing reports by specifying the physical appearance of a report, thereby relieving the programmer of having to specify the detailed procedures necessary to produce that report.

The purpose of this manual is to introduce the concept of reports and to describe those specific features of the COBOL language used to produce them.

In order to accomplish this, the manual is organized as follows:

- Section 1 contains a general introduction to applications and terminology.
- Section 2 presents Report Writer within the context of a COBOL program showing the major entries required for Report Writer.
- Section 3 illustrates and explains the usage and capabilities of Report Writer.
- Section 4 contains an example of a complete report and the Report Writer program that produced it.
- Appendix A presents the formats and rules for Report Writer.



## I. THE REPORT

### 1.1. GENERAL

Business, industrial, and government operations require organized information in the form of reports. Commonly used reports are for payroll, sales, and inventory control operations. The Report Writer feature facilitates writing a program for generating a report.

To produce a report using the COBOL language without the Report Writer feature requires the programmer to write coding to manipulate either input data or data created by the program, to construct a print line. Each line can then be printed by issuing a WRITE statement. In addition, the programmer has the responsibility of writing procedures for maintaining a page count, a line count for each page, for producing headings, summing data fields, and checking for changes in a given data field to know when to print subtotals and totals.

To produce a report using the Report Writer requires only three commands in the Procedure Division and a description of the report in the Data Division. In addition to the standard sections of the Data Division such as the File and Working-Storage Sections, a Report Section is required which contains the report description.

Transaction data to be included in the report can be created in the program or transferred to storage using an input statement. The printing of one or more lines of a report is accomplished by issuing a GENERATE statement. The WRITE statement is not used, and manipulation of data in the Procedure Division is not required.

Report Writer is easily incorporated into a COBOL program providing the COBOL compiler contains the Report Writer feature. It requires no changes in the procedure for writing a COBOL program, nor does it limit the operations that can be performed by a COBOL program.

Before presenting a detailed explanation of the Report Writer statements, an explanation of basic terminology for describing reports is discussed.

## 1.2. PARTS OF A REPORT

This section presents the terminology required to define the parts of a report. (Report Writer provides such terms as COLUMN NUMBER, LINE NUMBER, VALUE IS, and SOURCE IS indicating from where the data is to be taken in the File or Working-Storage Sections.) A discussion of how a report is written is the subject of Section 3.

There are many different formats a report may take depending on its purpose. A simple report format could consist merely of a listing of transactions. However, headings above a listing facilitate reading and understanding the report. Also, the programmer does not have to include headings in the report description when pre-printed forms are used.

## 1.2.1. Page Heading, Footing, and Detail Data

Three basic elements of a report are as follows. (The abbreviations shown in parentheses are acceptable for use with Report Writer):

- Page heading (PH)
- Detail (DE)
- Page footing (PF)

The *page heading* consists of descriptive information which is printed at the top of each page, after page overflow occurs. In addition to a page heading, a report can have a *page footing* which is printed only at the bottom of every page.

A detail line appears between page heading and page footing information and supplies current transaction information related to one or more items. The information may be obtained from a transaction file.

It is possible to omit the page heading and/or page footing and have only detail information listed if the reader of the report knows the meaning of the data in the columns or is provided with a key to interpret the output.

Figure 1-1 contains a sample report showing the use of page heading, detail lines, and page footing.

APRIL 1970						
CITY	DRYERS	RADIOS	TV SETS	WASHERS	TOTALS	PAGE HEADINGS
ALBANY	10,000	1,000	8,000	1,000	\$20,000	DETAILS
BUFFALO	15,000	1,200	10,000	1,000	\$27,200	
ELMIRA	10,000	1,000	8,000	1,000	\$20,000	
NEW YORK	25,000	1,800	15,000	2,000	\$43,800	
SYRACUSE	20,000	2,500	10,000	2,000	\$34,500	
PAGE 81						PAGE FOOTING

Figure 1-1. Basic Elements of a Report



## 1.2.2. Control Headings and Control Footings

The programmer may wish to have detail lines arranged into groups and bounded by a heading and a footing. Information printed before a set of detail lines is called a *control heading* (CH); information presented after a set of detail lines is called a *control footing* (CF). Control headings and footings contain descriptive information, but control footings can also be used to produce running totals of values that are automatically kept track of by the Report Writer.

Control headings and control footings are produced only when a *control break* occurs. A control break is a change in the normal processing of transaction data. The basis for the change is a field that contains the same information in each record of a group of records. The grouping of transactions may be according to a time period (the week or month), a geographical area (city, state), department, or employee number. The field chosen by the programmer is examined by the Report Writer prior to each printing operation. When the information in the field changes (from February to March; from Albany to Buffalo) a *control break* occurs. The control break results in the production of printed descriptive information (the control footing) pertaining to the previous set of detail lines, as well as information (the control heading) pertaining to the next set of detail lines.

In the example given in Figure 1-2, a change in the state caused the control break. When the state changed from New York to Penna., a control footing was printed for the New York transaction group, NEW YORK TOTALS ..., followed by a Penna. control heading, PENNA INCOME.

		APPLIANCE COMPANY – NATIONAL SALES							
		APRIL 1970							
PAGE HEADINGS		REGION	STATE	CITY	DRYERS	RADIOS	TV SETS	WASHERS	TOTALS
CONTROL HEADINGS		—NEW YORK INCOME							
DETAIL TRANSACTIONS		EASTERN	NEW YORK	ALBANY	10,000	1,000	8,000	1,000	\$ 20,000
		EASTERN	NEW YORK	BUFFALO	15,000	1,200	10,000	1,000	\$ 27,200
		EASTERN	NEW YORK	ELMIRA	10,000	1,000	8,000	1,000	\$ 20,000
		EASTERN	NEW YORK	NEW YORK	25,000	1,800	15,000	2,000	\$ 43,800
		EASTERN	NEW YORK	SYRACUSE	<u>20,000</u>	<u>2,500</u>	<u>10,000</u>	<u>2,000</u>	<u>\$ 34,500</u>
CONTROL FOOTING		—NEW YORK TOTALS			\$80,000	\$7,500	\$51,000	\$7,000	\$145,500
CONTROL HEADING		—PENNA INCOME							
		EASTERN	PENNA	ALLENTOWN	5,000	2,000	15,000	5,000	\$ 27,000

Figure 1-2. Control Headings and Control Footings

1.2.3. Report Heading and Report Footing

The remaining two parts of a report are the *report heading* (RH) and the *report footing* (RF). The report heading is printed only on the first page of the report. The report footing is printed only on the last page of the report. The report heading and report footing are illustrated in Figure 1-3 which shows all the parts of a report.

Assuming all the parts of a report are desired, the order in which the report groups are printed is as follows:

- Report Heading
- Page Heading
- Control Heading
- Detail
- Control Footing
- Page Footing
- Report Footing

APPLIANCE COMPANY								REPORT HEADING (first page only)
NATIONAL SALES REPORT								
APRIL 1970								
APPLIANCE COMPANY - NATIONAL SALES								PAGE HEADING
APRIL 1970								
REGION	STATE	CITY	AIR COND	RADIOS	TV SETS	WASHERS	TOTALS	
WASHINGTON INCOME								
WESTERN	WASHINGTON	SEATTLE	1,000	1,000	4,500	1,500	\$ 8,000	
		TACOMA	1,000	500	3,500	1,000	\$ 6,000	
		WASHINGTON TOTALS.....	\$ 2,000	\$ 1,500	\$ 8,000	\$ 2,500	\$ 14,000	
COLORADO INCOME								
WESTERN	COLORADO	BOULDER	6,000	7,000	8,000	5,000	\$ 26,000	
		COLO SP	5,000	6,000	4,000	8,000	\$ 23,000	
		DENVER	10,000	10,000	20,000	7,000	\$ 47,000	
		COLORADO TOTALS.....	\$ 21,000	\$ 23,000	\$ 32,000	\$ 20,000	\$ 96,000	
		WESTERN REGION TOTALS.....	\$ 59,200	\$ 57,000	\$ 102,000	\$ 50,000	\$ 268,200	
APRIL TOTAL SALES.....					\$ 665,600			

PAGE 08 ----- PAGE FOOTING

END OF REPORT --- REPORT FOOTING  
(last page only)

Figure 1-3. The Complete Report

## 2. REPORT WRITER IN A COBOL PROGRAM

### 2.1. GENERAL

This section illustrates the additions to a COBOL program required for describing a report.

### 2.2. REPORT WRITER ENTRIES

The following divisions of a COBOL program require information to produce a report using the Report Writer.

#### ■ ENVIRONMENT DIVISION

No additions are made to the Environment Division. However, the SELECT and ASSIGN statements are used to select the file-name and assign that file-name to the hardware device to which the report is to be transferred.

#### ■ DATA DIVISION

In addition to FDs for input and output files, the Report Writer requires an FD describing the output file which is to contain the report.

This FD specified for the Report Writer is different than the conventional FDs in two respects:

- (a) in addition to the clauses describing a file, the FD must contain a REPORT clause which can specify one or more report names and,
- (b) level indicators (01, 02, etc.) are not specified under the FD.

Following the FDs in the File Section is a Report Section. The Report Section contains an RD (Report Description) which specifies the name of the report. The RD also describes where the heading, footing, and detail information are to appear on a page and specifies the identifiers used to determine when a control break is to occur. The reader should note the difference between a *file-name* and a *report-name*. The file-name appears in the SELECT clause and in the corresponding FD, whereas the report-name appears in the REPORT clause (in the File Section) and in the corresponding RD (in the Report Section). The relationship between the names is shown in the following example:

ENVIRONMENT DIVISION.

SELECT **SALES** ASSIGN TO PRINTER. file-name SALES

DATA DIVISION.

FILE SECTION.

FD **SALES**

REPORT IS **INVENTORY**

report-name INVENTORY

REPORT SECTION.

RD **INVENTORY**

Following the RD are the 01 level report groups. A report group is similar to the 01 level record description following an FD and is identified at the 01 level by a TYPE IS clause. Each report group desired, heading, footing, and detail, requires a separate report group description.

#### ■ PROCEDURE DIVISION

Three new statements are required, INITIATE and TERMINATE, which are similar to OPEN and CLOSE statements except that they are used to initialize report operations and subsequently to terminate report operations. However, all files used by the Report Writer must be opened and closed by the program. The third statement, GENERATE, produces printing or causes summations to be performed without printing.

A fourth statement, USE BEFORE REPORTING, is included as an option. It provides a means of including information in the report that was not available until just before a GENERATE is issued. For example, a programmer may want a header to specify descriptive information which is dependent upon the current value of a line or page count.

Figure 2-1 illustrates the basic report language within a COBOL program.




Figure 2-1. Report Writer Source Coding

All files must be opened. Before any report lines can be printed, an INITIATE (report-name) statement must be executed. Data may then be transferred to storage by statements such as a READ, or data for the report may be created by the program.

When detail information is ready for printing, a GENERATE statement is issued containing the name associated with the detail report group, i.e., GENERATE PAY. However, if the programmer only wants summations of various fields to occur but does not want detail printing, he must specify the report name in the GENERATE statement.

When the end-of-file is detected, a TERMINATE report-name must be executed which causes the printing of any final totals and desired footings to be printed.

## 3. USING THE REPORT WRITER

### 3.1. GENERAL

This section describes the Data Division and the Procedure Division statements for producing a report. Appendix A formally describes the rules for usage as put forth by ANSI. Specific implementations of Report Writer may vary from the standard specifications.

### 3.2. DATA DIVISION ENTRIES

The coding format of a report begins with the section header, REPORT SECTION, followed by a Report Description (RD). The RD contains a PAGE clause describing the general physical layout of information for each page, and the CONTROL clause specifying identifiers naming the fields which initiate control breaks.

Beneath the RD is one or more report groups describing the parts of a report. A report group is a set of data items that is to be produced as a unit. It may consist of several report lines containing many data items or of one report line containing a single data item. Three categories of report group definitions are provided: heading groups, footing groups and detail groups. A separate report group is written for each kind of heading, footing, and detail transaction.

If more than one report is described in a COBOL program, a CODE clause is specified in order to label each report line for later selection. For instance, if more than one report is written to a tape unit, the mnemonic-name supplied with the CODE clause in the RD is used by a tape-to-printer routine to select the desired report material for output.

Most of the report is described by information specified in report groups. A report group, similar to a record description, can be described using various combinations of group and elementary items. The following generalized examples show some ways a report group can be described.

To present one line of printed information:

01 TYPE IS\_\_LINE\_\_COLUMN\_\_PIC\_\_VALUE/SOURCE/SUM\_\_.

To present one line with more than one value to be printed:

01 TYPE IS\_\_LINE\_\_.

02 COLUMN\_\_\_\_PIC\_\_\_\_VALUE/SOURCE/SUM\_\_\_\_.

02 COLUMN\_\_\_\_PIC\_\_\_\_VALUE/SOURCE/SUM\_\_\_\_.

To present more than one line of information:

01 TYPE IS\_\_.

02 LINE\_\_.

03 COLUMN\_\_\_\_PIC\_\_\_\_VALUE/SOURCE/SUM\_\_\_\_.

03 COLUMN\_\_\_\_PIC\_\_\_\_VALUE/SOURCE/SUM\_\_\_\_.

02 LINE\_\_\_\_.

03 COLUMN\_\_\_\_PIC\_\_\_\_VALUE/SOURCE/SUM\_\_\_\_.

03 COLUMN\_\_\_\_PIC\_\_\_\_VALUE/SOURCE/SUM\_\_\_\_.

02 LINE\_\_COLUMN\_\_PIC\_\_VALUE/SOURCE/SUM\_\_\_\_.

3.2.1. Report Description (RD)

Figure 3-1 and the accompanying table illustrates how an RD is written. The entries comprising a clause could be written on the same line; semicolons are optional. The rules for coding are the same as those used for writing a COBOL program.

		A	B	TEXT					
6	7	8	11	12	20	30	40	50	60
				RD	SALES ;	CODE	SALES ;		
					PAGE	LIMIT	IS	60 ;	
					HEADING	10 ;			
					FIRST	DETAIL	20 ;		
					LAST	DETAIL	55 ;		
					FOOTING	58 ;			
					CONTROLS	ARE	REGION ;	STATE ;	
				01	TYPE	IS	REPORT	HEADING	

Figure 3-1. Sample RD



CAUSE	EXPLANATION
RD SALES	SALES is the name of the report. The name must appear in the REPORT IS clause of the FD associated with the report. The name can be referenced in a GENERATE statement which will result in summary reporting (detail lines will not be printed, only headings and footings).
PAGE LIMIT IS 60	Indicates that the last line that can be printed is line 60. The line number does not have to be the same as the physical number of lines that can appear on a page.
HEADING 10	No report group will be printed above line 10.
FIRST DETAIL 20	This detail specification refers to both detail transactions and control headings. Neither a detail line nor a control heading will be printed above line 20.
LAST DETAIL 55	A detail line will not be printed after line 55.
FOOTING 58	A control footing will not be printed after line 58.
CODE SALES	SALES is the name of the report used by an I/O routine to select the report from a number of reports that are on the input file.
CONTROL	REGION, STATE, indicate that a control break is to occur when a change in the region or the state occurs. The sequence in which the names are specified implies a hierarchical order.

3.2.2. Sample Report Group Description Entry

A report group, similar to a record description, contains all the information associated with an 01 level. Unlike a record description for a printer line, a report group can describe information for more than one printed line. Although there are seven types of report groups (RH, PH, CH, DE, DF, PF, and RF), a basic format is used to describe the essential features of any report group. A few additions are used with the detail and control footing report groups; these are discussed in the paragraphs describing the specific groups.

An example of a report group for printing a line such as

SUM IS \$1,856

is as follows:

A	B	TEXT
6	7	8 11 12 20 30 40 50 60
01		SOLD; TYPE IS DETAIL; LINE NUMBER IS PLUS
02		COLUMN 5, PIC X(7), VALUE IS 'SUM IS'
02		COLUMN 2, PIC \$9,999, SOURCE IS COUNTY-ER.
		GENERATE SOLD.

The basic entries comprising a report group are shown in Table 3-1.


REPORT GROUP ENTRY	EXPLANATION	EXAMPLE
TYPE IS	Specifies type of report group.	TYPE IS CONTROL HEADING
LINE NUMBER IS	Specifies line number to be printed on.	LINE NUMBER IS 10 or LINE NUMBER IS PLUS 1 (relative)
COLUMN NUMBER IS	Specifies the column number where the first printed character is to appear.	COLUMN NUMBER IS 15
PICTURE	Describes the data type and size of field.	PICTURE X(20) PICTURE \$\$Z,999
VALUE IS	Describes a constant to be printed.	VALUE IS 'TOTAL PARTS SOLD'
SOURCE IS	Specifies an identifier described in an FD or a Working-Storage entry whose value is to be printed.	SOURCE IS SOC-SEC-NO

Table 3-1. Report Group Entries

The following paragraphs describe each type of report group.

### 3.3. THE HEADING AND FOOTING

Report group coding for report and page headings, and report and page footings is essentially the same except that the TYPE IS clause specifically defines the kind of report group. Assume a page of a report is to contain a report heading (first page of a report only) and a page heading (top of each report page). Figure 3-2 shows the relationship between the coding of report groups and the printed lines on a report.



LINE	A	B	TEXT	20	30	40	50	60
0.1			TYPE IS REPORT HEADING.					
0.2			LINE NUMBER IS 10; COLUMN NUMBER IS 49					
			PIC X(14) VALUE IS 'XYZ APPLIANCES'					
0.2			LINE NUMBER IS 11; COLUMN NUMBER IS 41					
			PIC X(21) VALUE IS 'NATIONAL SALES REPORT'					
0.2			LINE NUMBER IS 12					
			03 COLUMN NUMBER IS 49; PIC X(8) SOURCE IS MONTH					
			03 COLUMN NUMBER IS 55; PIC X(4) VALUE IS '1970'					
0.1			TYPE IS PAGE HEADING.					
0.2			LINE NUMBER IS 15; COLUMN NUMBER IS 33					
			PIC X(31) VALUE 'XYZ APPLIANCES - NATIONAL SALES'					
0.2			LINE NUMBER IS 16					
			03 COLUMN NUMBER IS 49; PIC X(8) SOURCE IS MONTH					
			03 COLUMN NUMBER IS 58; PIC X(4) VALUE IS '1970'					

Figure 3-2. Report Group and Printed Line Relationship

This figure shows the first page of a report where the report heading and the page headings are printed on that page. If the programmer does not wish the page heading to be printed on the page with the report heading, the NEXT GROUP clause can be specified at the 01 level of the report heading report group. Thus, the first line of the report heading would be written as follows:

```
01 TYPE IS REPORT HEADING; NEXT GROUP IS NEXT PAGE.
   02 LINE 10 COLUMN . . .
```

The page heading is printed on line 15 of the second page as specified in the PH report group; lines 10 through 14 are blank.

Assume that a heading is to be printed on line 10 regardless of the type of heading. The RD and report groups in the following example produce a report heading printed on line 10 of the first page, and a page heading on line 10 of every successive page.

Headings are printed on the line number which results from the addition of the value specified in the HEADING clause (9) and the value indicated in the report group entry (PLUS 1). Thus, the RD is printed on line 10 of the first page; the next group, PH, is printed on line 10 of each succeeding page.

The line number specified in a heading report group must be equal to or greater than the line number specified in the HEADING clause of the RD.

A	B	TEXT
6	7	8
11	12	20
		30
		40
		50
		60
RD		ANNUAL
		PAGE LIMIT IS 60
		HEADING 91
		:
01		TYPE IS REPORT HEADING LINE NUMBER IS PLUS 1
		NEXT GROUP IS NEXT PAGE
		:
01		TYPE IS PAGE HEADING LINE NUMBER IS PLUS 1

In the two preceding examples, three ways of designating a line number or page have been used:

- LINE NUMBER IS 10
- LINE NUMBER IS PLUS 1
- LINE NUMBER IS NEXT PAGE

In (a), line number is a positive integer. In (b), line number is relative to the value in the line counter maintained by the Report Writer. When the first line of the page is to be printed, the line number is 1 plus the value specified in the HEADING clause; when a line other than the first is to be printed, the line number is 1 plus the previous value in the line counter.

To skip to the top of the following page, the key words `NEXT PAGE` are used, as shown in (c).

The options specifying the line number are the same for the `NEXT GROUP` clause. However, the `LINE NUMBER` clause can be used at any level whereas the `NEXT GROUP` clause can only be specified at a 01 level.

The general form for specifying line values is as follows:

$$\left[ \text{; } \underline{\text{LINE NUMBER}} \text{ IS } \left\{ \begin{array}{l} \textit{integer-1} \\ \underline{\text{PLUS } \textit{integer-2}} \\ \underline{\text{NEXT PAGE}} \end{array} \right\} \right]$$

$$\left[ \text{; } \underline{\text{NEXT GROUP}} \text{ IS } \left\{ \begin{array}{l} \textit{integer-3} \\ \underline{\text{PLUS } \textit{integer-4}} \\ \underline{\text{NEXT PAGE}} \end{array} \right\} \right]$$

#### 3.4. THE BODY OF A REPORT PAGE

The *body* of a report refers to the area on a page that is to contain the detail lines, control headings, and control footings. Control headings and control footings mark off groups of detail lines within the body area of each page. The number of lines in the body area is defined by information stated in the `PAGE` clause. For example:

```
RD TEST;
  PAGE LIMIT IS 60;
  HEADING 5;
  FIRST DETAIL 20;
  LAST DETAIL 40;
  FOOTING 50;
```

The `PAGE` clause indicates that:

FIRST DETAIL 20            No line of a detail or control group will start before line 20.

LAST DETAIL 40            No line of a detail or control heading group will be printed beyond line 40.

FOOTING 50                No control footing print group will be printed beyond line 50.

Thus, the `PAGE` clause indicates that a transaction detail line will not extend beyond line 40 although a control footing print group can be printed up to and including line 50.

A page limit of 60 means that page and report footings can be printed up to and including line 60.

The general format of the PAGE clause is as follows:

$$\left[ ; \text{PAGE} \left[ \begin{array}{l} \text{LIMIT IS} \\ \text{LIMITS ARE} \end{array} \right] \text{integer-1} \left\{ \begin{array}{l} \text{LINE} \\ \text{LINES} \end{array} \right\} [ , \text{HEADING integer-2}] \right. \\ \left. [ , \text{FIRST DETAIL integer-3}] [ , \text{LAST DETAIL integer-4}] \right. \\ \left. [ , \text{FOOTING integer-5}] \right]$$

### 3.4.1. The Detail Report Group

The entries comprising a detail report group are the same as those previously described for headings and footings. In addition, the GROUP INDICATE clause is specified only in a detail report group. It is used to suppress unnecessary repetitive information described at an elementary item level.

#### 3.4.1.1. Multiple Detail Report Groups

More than one detail report group may be written in a report description. Each report group can contain a SOURCE clause that specifies the same identifier. When the identifier is used in a summing operation, the UPON clause is required to indicate which detail report group is involved in the summation. For example:

RD PAGE clause

CONTROL WH-1, WH-2.

01 WAREHSE-1 TYPE IS DETAIL.

02 LINE PLUS 1 COLUMN 15 PIC 9(4)

SOURCE PARTS.

01 WAREHSE-2 TYPE IS DETAIL.

02 LINE PLUS 1 COLUMN 15 PIC 9(4)

SOURCE PARTS.

01 TYPE IS CONTROL FOOTING WH-1.

02 LINE PLUS 2.

03 COLUMN 10 PIC X(11) VALUE 'TOTAL PARTS'.

03 COLUMN 25 PIC 9(4) SUM PARTS UPON WAREHSE-1.

01 TYPE IS CONTROL FOOTING WH-2.

02 LINE PLUS 2

03 COLUMN 10 PIC X(11) VALUE 'TOTAL PARTS'.

03 COLUMN 25 PIC 9(4) SUM PARTS UPON WAREHSE-2.

When the statement GENERATE WAREHSE-1 is executed, the current value of source item PARTS is added into a sum counter associated with the WAREHSE-1 detail report group; similarly, when the statement GENERATE WAREHSE-2 is executed, the current value of source item PARTS is added to a sum counter associated with the WAREHSE-2 detail report group. A control break resulting from a change in WH-1 results in the printing of the sum counter value of WAREHSE-1; a control break resulting from a change in WH-2 results in printing of the sum counter value of WAREHSE-2.

Since neither of the elementary items containing a sum clause is referenced, the item is not provided with a data name following the 03 level indicator.

### 3.4.1.2. Group Indicate

The value associated with an item that is associated with a GROUP INDICATE clause is printed the first time the value occurs. The value is not printed again until either a control break or a top-of-page condition occurs.

A group of records may have information common to all records. For instance, each employee's weekly payroll record contains the department and division number. Each line printed, representing one transaction, repeats the department and division number. To eliminate unnecessary repetitive information, which incidentally improves the appearance and readability of a report, a GROUP INDICATE clause is included in the entry of each elementary item describing the department and division number. Figure 3-3 illustrates a report both with and without the use of the GROUP INDICATE clause. (The page heading coding shown is the same for both examples.)

The NEXT GROUP clause in the PH report group indicates two lines are to be skipped before the detail line is printed.

The detail line specifies LINE PLUS 1 indicating an additional line is to be skipped before printing. Thus, the first detail line is printed on the third line after the page heading; each succeeding line is single spaced.

The region and state information are not repeated after the first detail line because the cities listed are in the same state and region. WESTERN and OREGON are printed because a control break occurred for both region and state. The information required to describe a control break, not shown in the example, is discussed in the following section.

### 3.5. CONTROL HEADING AND FOOTING REPORT GROUPS

Both control headings and footings are printed when a control break occurs. A control break is initiated by a change in the value associated with an identifier specified after the TYPE IS clause. The identifier must appear in the CONTROLS clause of the RD.

Example:

```
RD SALES
CONTROLS ARE DIVISION, DEPT.
01 TYPE IS CONTROL HEADING DEPT.
.
.
.
01 TYPE IS CONTROL FOOTING DIVISION.
.
```

PAGE HEADER

A	B	TEXT
01		TYPE IS PH; LINE 20; NEXT GROUP IS PLUS 2.
02		COLUMN 2, PIC X(6), VALUE 'REGION'.
02		COLUMN 12, PIC X(5), VALUE 'STATE'.
02		COLUMN 25, PIC X(4), VALUE 'CITY'.
02		COLUMN 40, PIC X(5), VALUE 'AIR COND'.
02		COLUMN 55, PIC X(6), VALUE 'RADIOS'.

REGION AND STATE WITHOUT GROUP INDICATE

REGION	STATE	CITY	AIR COND	RADIOS
EASTERN	NEW YORK	ALBANY	10,000	1,000
EASTERN	NEW YORK	BUFFALO	15,000	1,200
EASTERN	NEW YORK	ELMIRA	10,000	1,000
EASTERN	NEW YORK	NEW YORK	25,000	1,800
EASTERN	NEW YORK	SYRACUSE	20,000	2,500
Western	Oregon	Eugene	1,200	2,500
Western	Oregon	Portland	1,500	2,000

UNNECESSARY REPETITIVE DATA

TRANS	TYPE	IS	DETAILS	LINE	PLUS
01					
02					
02					
02					
02					

REGION AND STATE WITH GROUP INDICATE

REGION	STATE	CITY	AIR COND	RADIOS
EASTERN	NEW YORK	ALBANY	10,000	1,000
		BUFFALO	15,000	1,200
		ELMIRA	10,000	1,000
		NEW YORK	25,000	1,800
		SYRACUSE	20,000	2,500
WESTERN	OREGON	EUGENE	1,200	2,500
		PORTLAND	1,500	2,000

TRANS	TYPE	IS	DETAILS	LINE	PLUS
01					
02					
02					
02					
02					

Figure 3-3. Using the GROUP INDICATE Clause



When the department number changes, information concerning the department is printed preceding a group of detail transactions. When the division number changes, information is printed after the report groups required for that division.

Thus, one difference between control report groups and other report groups is the necessity of a control break identifier. Although the control heading and footings can contain the same entries as other control groups, the control footing report groups can specify a clause to permit automatic summing of selected values.

When more than one control heading or control footing is specified, the order in which multiple headings and footings are printed is determined by two factors:

- (a) the type of control group, i.e., heading or footing and
- (b) the sequence of the identifiers as they appear in the CONTROL clause.

For example, if a CONTROL clause appears as follows:

CONTROLS ARE YEAR, MONTH, WEEK.

and three *control headings* are specified, a control break caused by a change in the year results in three control headings to be printed in the following order:

year  
month  
week

If three *control footings* are specified, the order in which the footings are printed is reversed:

week  
month  
year

### 3.5.1. Summing Methods

Two methods of forming sums are called *running totals* and *crossfooting totals*. The running total method is the process of successive addition of values which are written in columnar form. The final sum of a set of values is called a *sum total*, however, if more than one such sum exists where each is to be added to another, each sum is then referred to as a *subtotal*. The addition of one subtotal to another subtotal is called *rolling totals*. Rolling totals are created during control breaks when a subtotal is added to a previously formed subtotal.

Where a running total is produced by adding items in the same column reading down a page, a crossfooting total (as used in Report Writer) is produced by adding running totals that form a row across a page. Figure 3-4 illustrates running and crossfooting sums and the coding used to produce them.

JANUARY	PONTIAC	CHEVROLET		
011070	0058	0150		
011770	0100	0161		
013170	<u>0292</u>	<u>0321</u>		
<b>RUNNING TOTALS</b>	0450	0632	TOTAL \$1,082	<b>CROSSFOOT TOTAL</b>

```

01 SALES-REPORT; TYPE IS DE; LINE PLUS 1;
02 COLUMN 20 PIC 9(4) SOURCE PONTIAC.
02 COLUMN 30 PIC 9(4) SOURCE CHEVROLET.

01 TYPE IS CF MONTH.
02 LINE PLUS 1.
03 COLUMN 20 PIC X(4) VALUE '----'.
03 COLUMN 30 PIC X(4) VALUE '----'.
02 LINE PLUS 1.
03 PNT COLUMN 20 PIC 9(4) SUM PONTIAC.
03 CHV COLUMN 30 PIC 9(4) SUM CHEVROLET.
03 COLUMN 45 PIC X(5) VALUE 'TOTAL',
03 COLUMN 52 PIC $Z,Z99 SUM PNT, CHV.

```

Figure 3-4. Running and Crossfoot Sums

A crossfoot sum is specified by following the SUM clause with the item names within the CF report group. PNT and CHV (see Figure 3-4) are the names of fields which the Report Writer uses to sum the values. Summing the two fields gives a crossfoot total. When a single value is required, the names in the detail group are referenced in the CF, e.g., SUM PONTIAC in the CF references SOURCE PONTIAC in the detail group.

The following illustration shows the summing method. Heading and descriptive information is omitted for clarity. Assume the CF report groups, providing for three control breaks, will print the total payroll for:

- each department
- each division
- the company

Counter names appear after the 01 level designation. Three counters will be provided by the Report Writer:

- DEPT-PAY maintains a running total of each employee's payroll in the department.
- DIV-PAY maintains a rolling total of the payroll for an entire division.
- TOT-PAY maintains a rolling total of the payroll for the company. This counter name can be omitted because it is not referenced in any other entry. The Report Writer automatically provides a location for this counter.

Data-name MONY is referenced in the GENERATE statement to generate printing.



7	8	9	10	11	12	13	14	15	16	17	18	19	20	21	22	23	24	25	26	27	28	29	30	31	32	33	34	35	36	37	38	39	40	41	42	43	44	45	46	47	48	49	50	51	52	53	54	55	56	57	58	59	60
JANUARY										PONTIAC										CHEVROLET																																	
011070										0050										0120																																	
011770										0100										0050																																	
013170										0232										0321																																	
										0450										0632										TOTAL SALES \$1082																							
FEBRUARY										PONTIAC										CHEVROLET																																	
020770										0035										0075																																	
022970										0102										0085																																	
										0500										0300										TOTAL SALES \$0800																							
																														TOTAL SALES AMOUNT \$1,882																							

FD CARS LABEL RECORD IS OMITTED DATA RECORD IS TRANS.

01 TRANS.


02 MONTH PIC X(8).

02 WK-END PIC 9(6).

02 PONT PIC 9(4).

02 CHEV PIC 9(4).

02 FILL PIC X(58).



FD FOR-REPORT LABEL RECORDS ARE OMITTED; REPORT IS CAR-SALES.  
REPORT SECTION.

RD CAR-SALES; CONTROL IS FINAL, MONTH  
PAGE LIMIT IS 60.

01 TYPE IS CH MONTH LINE 20.

02 COLUMN 10 PIC X(8) SOURCE IS MONTH.

02 COLUMN 20 PIC X(7) VALUE IS 'PONTIAC'.

02 COLUMN 30 PIC X(9) VALUE IS 'CHEVROLET'.

01 PRINT; TYPE IS DE; LINE PLUS 1.

02 COLUMN 10 PIC 9(6) SOURCE WK-END.

02 COLUMN 20 PIC 9(4) SOURCE PONT.

02 COLUMN 30 PIC 9(4) SOURCE CHEV.

01 TYPE IS CF MONTH.

02 LINE PLUS 1.

03 COLUMN 20 PIC X(4) VALUE '----'.

03 COLUMN 30 PIC X(4) VALUE '----'.

02 LINE PLUS 1.

03 PONTIAC COLUMN 20 PIC 9(4) SUM PONT.

03 CHEVLET COLUMN 30 PIC 9(4) SUM CHEV.

03 COLUMN 42 PIC X(11) VALUE IS 'TOTAL SALES'.

03 CRS-TOT COLUMN 56 PIC \$9999 SUM PONTIAC, CHEVLET.

CROSS FOOT  
TOTAL

01 TYPE IS CF FINAL.

02 LINE PLUS 4 COLUMN 10 PIC X(18) VALUE IS 'TOTAL SALES AMOUNT'.

02 LINE PLUS 1 COLUMN 30 PIC \$9,999 SUM CRS-TOT.

Figure 3-6 Sample Report

### 3.5.2. The Sum Counter Reset Option

The programmer may not wish to have a particular sum counter reset to zero after a control break. That is, after a control footing has been printed, it may be desirable to continue adding into the same sum counter. For example, a control footing is written which results in a control break when the department number changes. Thus, the sum printed reflects a cumulative total for all departments rather than a separate total for each department.

To inhibit the reset after a control break, the programmer must specify a RESET clause in the sum counter item entry. The RESET clause contains an identifier (or the reserved word FINAL) specifying the name of a higher level control indicating when the counter is to be reset to zero. The entry appears as follows:

```
01 TYPE IS CF DEPT LINE PLUS 2.  
02 DSUM COLUMN 10 PIC 9(4) SUM SALARY RESET DIV .  
01 TYPE IS CF DIV LINE PLUS 2.
```

The counter DSUM is not reset to zero until the division number changes.

### 3.6. FIXED DATA NAMES

Two fixed data-names, LINE-COUNTER and PAGE-COUNTER, can be referenced as source items in a report group or referenced in Procedure Division statements. The counters are provided by the Report Writer when the counter name is specified as a source item. The PICTURE clause associated with the item defines the size of the counter.

The line counter location is set to 0 and the page counter location is set to 1 when the INITIATE statement is executed.

A typical use of the page counter is for printing report page numbers. A report group for this purpose appears as follows:

```
01 TYPE IS PAGE FOOTING LINE 52 .  
02 COLUMN 38 PIC X(4) VALUE 'PAGE'.  
02 COLUMN 43 PIC 99 SOURCE PAGE-COUNTER.
```

A line counter is provided by the Report Writer only when the PAGE LIMIT clause is included in the Report Description (RD).

### 3.7. PROCEDURE DIVISION

The four statements pertaining to Report Writer are INITIATE, GENERATE, TERMINATE and USE BEFORE REPORTING.

The INITIATE statement specifies one or more report-names and is required before processing a report.

The GENERATE statement specifies either a report-name (called summary reporting) or a data-name (for detail reporting) associated with a detail report group. When a report-name is used, detail printing does not occur, only summing operations are performed.

The TERMINATE statement specifies the report-name. It is required to terminate the report and results in printing the last set of Control Footing report groups and the Report Footing.

The USE BEFORE REPORTING is used when information required for a report is available only immediately before printing a report group. This statement can only be used in the DECLARATIVES in the Procedure Division. (An explanation of DECLARATIVES is given in the Fundamentals of COBOL Language UP-7503.1 (current version).)

When a GENERATE verb is issued, and Declarative procedures have been specified for the report group, control is passed to the Declarative section for execution of the stated procedures. When the procedures have been executed, the execution of the GENERATE verb is resumed. Figure 3-7 shows that information is to be inserted into a page heading or page footing report group depending on the contents of a location called CBL-CTR.

```
PROCEDURE DIVISION.  
DECLARATIVES.  
RW-1 SECTION. USE BEFORE REPORTING PAGE-HED.  
RW-2. IF CBL-CTR IS LESS THAN 1 OR EQUAL TO 1  
    MOVE 'BEGIN' TO BEGIN-FIELD-FOR-PH  
    MOVE SPACES TO CONTINUED-FIELD-FOR-PH  
    ELSE MOVE SPACES TO BEGIN-FIELD-FOR-PH  
    MOVE 'CONTINUED' TO CONTINUED-FIELD-FOR-PH.  
RW-3 SECTION. USE BEFORE REPORTING PAGE-FOOT.  
RW-4. IF CBL-CTR IS GREATER THAN 1  
    MOVE MONTHNAME (MONTH) TO MONTH-FOR-PF  
    MOVE 'CONTINUED ON NEXT PAGE' TO TEXT-FOR-PF  
    ELSE MOVE SPACES TO MONTH-FOR-PF  
    MOVE SPACES TO TEXT-FOR-PF.  
END DECLARATIVES.  
MAIN-SECTION SECTION.  
START.  
    OPEN INPUT INFILE, OUTPUT REPORT-FILE.  
    INITIATE EXPENSE-REPORT.  
    MOVE 1 TO CBL-CTR.  
READATA, READ INFILE AT END GO TO COMPLETE.  
    GENERATE DETAIL-LINE. GO TO READATA.  
COMPLETE. TERMINATE EXPENSE-REPORT.  
    CLOSE INFILE, REPORT-FILE, STOP RUN.
```

Figure 3-7. Use of Report Writer Verbs

## 4. SAMPLE PROGRAM

### 4.1. GENERAL

This section shows the relationship between a report and the Report Writer program that produced it. The report and the program are shown in Figures 4-1 and 4-2.

### 4.2. THE PROGRAM

The following illustration shows the COBOL part of the program that precedes the Report Section. INPUT-DATA is the name of the input file containing the records to be processed. OUT-PUT is the name of the output file containing the name of the report (SALES).

```
IDENTIFICATION DIVISION.  
PROGRAM-ID. SALES REPORT.  
AUTHOR. AB. C. DEFGHI.  
DATE-WRITTEN. APRIL 70.  
ENVIRONMENT DIVISION.  
CONFIGURATION SECTION.  
SOURCE-COMPUTER. MARK-I.  
OBJECT-COMPUTER. MARK-I.  
INPUT-OUTPUT SECTION.  
FILE-CONTROL.  
    SELECT INPUT-DATA ASSIGN TO A CARD-READER.  
    SELECT OUT-PUT ASSIGN TO B PRINTER.  
DATA DIVISION.  
FILE SECTION.  
FD INPUT-DATA BLOCK CONTAINS 20 RECORDS LABEL RECORD IS OMITTED  
   DATA RECORD IS RE-CORD.  
01 RE-CORD.  
   02 REGION PIC X(8).  
   02 STATE PIC X(8).  
   02 CITY PIC X(8).  
   02 DESKS PIC 9(4).  
   02 CHAIRS PIC 9(4).  
   02 FILE-S PIC 9(4).  
02 FILLEP PIC X(44).  
FD OUT-PUT LABEL RECORD IS OMITTED REPORT IS SALES.  
WORKING-STORAGE SECTION.  
77 DETOT PIC 99999.
```

DETOT, defined in the Working-Storage Section, represents the name of a location for accumulating a crossfooting total for each detail line printed. The name appears in the Procedure Division as follows:

#### ADD DESKS, CHAIRS, FILE-S GIVING DETOT.

Recall that automatic crossfooting occurs only during a control break initiated by a control footing report group. The report entries follow the Working-Storage Section. The first report entry must be the words REPORT SECTION, which is followed by the RD:

#### REPORT SECTION.

```
RD SALES
   CONTROLS ARE FINAL, REGION, STATE
   PAGE LIMIT IS 60 LINES
   HEADING 4, FIRST DETAIL 10, LAST DETAIL 40, FOOTING 50.
```

FINAL in the CONTROLS clause is meaningful only in relation to a control footing. When the value associated with STATE changes, a STATE control break occurs. When the value associated with REGION changes, a STATE control break is executed followed by a REGION control break. At the termination of the program, the sequence is from minor control break to major control break, i.e., STATE, REGION, and FINAL.

Control headings are produced in the opposite sequence; major to minor. When a change occurs in the value associated with REGION, the Control Heading Region is printed first, followed by the heading for state.

The PAGE clause specifies the boundaries within the page for printing the various groups.

Following the RD are the report groups. Figures 4-1 and 4-2 illustrate a set of report group descriptions and the printed output. The report illustrates a report heading page and the last page of the report (page 03). This page contains the FINAL control break (producing 'NATIONAL TOTALS...'), and the report footing ('END OF REPORT'). Page two of the report, which is omitted, is similar except for the last two items mentioned which appears on page 03.

The encircled alphabetic characters in Figure 4-2 correspond to the encircled letters in Figure 4-1 which identify the report groups. The report groups for Oregon, Washington, and Colorado are not identified because they are essentially the same as the printed California report information.



```

01 TYPE IS REPORT HEADING NEXT GROUP IS NEXT PAGE.
02 LINE 5 COLUMN 32 VALUE IS 'XYZ OFFICE SUPPLIES'
   PIC X(19).
02 LINE 6 COLUMN 35 PIC X(14) VALUE 'NATIONAL SALES'.
02 LINE 7 COLUMN 36 PIC X(10) VALUE 'APRIL 1970'.
02 LINE 35 COLUMN 31 PIC X(19) VALUE 'HEADQUARTERS OFFICE'. (A)
02 LINE 36 COLUMN 31 PIC X(19) VALUE '102 TEDROCKA AVENUE'.
02 LINE 37 COLUMN 31 PIC X(19) VALUE 'PHILADELPHIA 20, PA'.
01 TYPE IS PAGE HEADING.
02 LINE PLUS 1 COLUMN 32 PIC X(19) VALUE 'XYZ OFFICE SUPPLIES'
   ' '.
02 LINE PLUS 1 COLUMN 36 PIC X(10) VALUE 'APRIL 1970'.
02 LINE PLUS 3.
03 COLUMN 41 PIC X(5) VALUE 'DESKS'. (B)
03 COLUMN 50 PIC X(6) VALUE 'CHAIRS'.
03 COLUMN 64 PIC X(5) VALUE 'FILES'.
03 COLUMN 75 PIC X(6) VALUE 'TOTALS'.
01 TYPE IS CONTROL HEADING REGION.
02 LINE PLUS 2.
03 COLUMN 2 PIC X(8) SOURCE REGION. (C)
03 COLUMN 10 PIC X(6) VALUE 'REGION'.
02 LINE PLUS 1 COLUMN 1 PIC X VALUE ' '.
01 LINE-OUT TYPE IS DE LINE PLUS 1.
02 COLUMN 20 PIC X(8) SOURCE STATE GROUP INDICATE.
02 COLUMN 29 PIC X(8) SOURCE CITY.
02 COLUMN 41 PIC Z,999 SOURCE DESKS. (D)
02 COLUMN 51 PIC Z,9(3) SOURCE CHAIRS.
02 COLUMN 63 PIC Z,9(3) SOURCE FILE-S.
02 COLUMN 75 PIC Z7,999 SOURCE DETOT.
01 TYPE IS CONTROL FOOTING STATE NEXT GROUP PLUS 1.
02 LINE PLUS 1 COLUMN 41 PIC X(5) VALUE '-----'.
02 COLUMN 51 PIC X(5) VALUE '-----'.
02 COLUMN 63 PIC X(5) VALUE '-----'.
02 COLUMN 75 PIC X(6) VALUE '-----'. (E)
02 LINE PLUS 1.
03 COLUMN 17 PIC X(8) SOURCE IS STATE.
03 COLUMN 26 PIC X(13) VALUE IS 'TOTALS.....'.
03 CFDSK COLUMN 39 PIC $Z,9(3) SUM DESKS.
03 CFCHR COLUMN 49 PIC $Z7,999 SUM CHAIRS.
03 CFFLS COLUMN 61 PIC $Z7,999 SUM FILE-S.
03 CFTOT COLUMN 73 PIC $Z7Z,9(3) SUM CFDSK, CFCHR, CFFLS.
01 TYPE IS CONTROL FOOTING REGION NEXT GROUP NEXT PAGE.
02 LINE PLUS 2.
03 COLUMN 1 PIC Y(8) SOURCE REGION.
03 COLUMN 9 PIC X(30) VALUE 'REGION TOTALS.....'
   '...'.
03 FDSK COLUMN 39 PIC $79,9(3) SUM CFDSK.
03 FCHR COLUMN 50 PIC $79,9(3) SUM CFCHR. (F)
03 FFLS COLUMN 62 PIC $79,9(3) SUM CFFLS.
03 SUBCROS COLUMN 73 PIC $799,9(3) SUM CFTOT.
02 TOT-CROSS PIC $Z99,999 SUM FDSK FCHR FFLS.
01 TYPE IS CONTROL FOOTING FINAL.
02 LINE PLUS 4.
03 COLUMN 1 PIC X(36) VALUE 'NATIONAL TOTALS FOR APRIL ..'
   '.....'.
03 COLUMN 38 PIC $799,999 SUM FDSK.
03 COLUMN 49 PIC $Z99,999 SUM FCHR. (G)
03 COLUMN 61 PIC $Z99,999 SUM FFLS.
03 COLUMN 71 PIC $Z,999,999 SUM TOT-CROSS.
01 TYPE IS PAGE FOOTING LINE 52.
02 COLUMN 38 PIC X(5) VALUE 'PAGE'. (H)
02 COLUMN 43 PIC 99 SOURCE PAGE-COUNTER.
01 TYPE IS REPORT FOOTING LINE 54.
02 COLUMN 36 PIC X(13) VALUE 'END OF REPORT'. (I)

```

Figure 4-1. Report Writer Program

XYZ OFFICE SUPPLIES  
NATIONAL SALES  
APRIL 1970

Ⓐ

HEADQUARTERS OFFICE  
102 TEDROCKA AVENUE  
PHILADELPHIA 20, PAXYZ OFFICE SUPPLIES  
APRIL 1970

Ⓑ

			DESKS	CHAIRS	FILES	TOTALS
Ⓒ	WESTERN REGION					
	CALIF	MONTEREY	2,000	100	750	2,850
Ⓓ		LOS ANGL	9,000	650	600	10,250
		SAN FRAN	2,000	150	800	2,950
		SAN JOSE	350	1,050	200	1,600
Ⓔ	CALIF	TOTALS.....	\$13,350	\$ 1,950	\$ 2,350	\$ 17,650
	OREGON	EUGENE	700	120	250	1,070
		PORTLAND	600	1,200	200	2,000
	OREGON	TOTALS.....	\$ 1,300	\$ 1,320	\$ 450	\$ 3,070
	WASHGTM	SEATTLE	500	250	150	900
		TACOMA	500	500	200	1,200
	WASHGTM	TOTALS.....	\$ 1,000	\$ 750	\$ 350	\$ 2,100
	COLORADO	BOULDER	3,000	3,000	1,250	7,250
		COLD SP	2,500	800	1,750	5,050
		DENVER	2,800	380	500	3,680
	COLORADO	TOTALS.....	\$ 8,300	\$ 4,180	\$ 3,500	\$ 15,980
Ⓕ	WESTERN REGION TOTALS.....\$23,950 \$ 8,200 \$ 6,650 \$ 38,800					
Ⓖ	NATIONAL TOTALS FOR APRIL ..... \$ 75,775 \$ 15,480 \$ 14,975 \$ 106,230					
Ⓗ	PAGE 03					
Ⓘ	END OF REPORT					

Figure 4-2. The Report

The Procedure Division to print the report is as follows:

PROCEDURE DIVISION.

START.

OPEN INPUT INPUT-DATA OUTPUT OUT-PUT.  
INITIATE SALES.

LOOP.

READ INPUT-DATA AT END GO TO FINIS.  
ADD DESKS, CHAIRS, FILE-S GIVING DETOT.  
GENERATE LINE-OUT.  
GO TO LOOP.

FINIS.

TERMINATE SALES.  
CLOSE INPUT-DATA OUT-PUT.  
STOP RUN.



## APPENDIX A. REPORT WRITER FORMATS

### A.1. GENERAL

The appendix is intended as a reference section describing the formats and rules for using the Report Writer statements.

### A.2. DATA DIVISION

The use of the Report Writer requires an FD section describing the characteristics of the file upon which the report will be written. The FD is not followed by (01 level indicator ) record descriptions.

In addition, a Report Section is required containing the entries describing the report. More than one report may be described.

#### A.2.1. File Description

##### Format:

FD *file-name*

[ ; BLOCK CONTAINS [*integer-1* TO *integer-2* { RECORDS / CHARACTERS } ]

; LABEL { RECORD IS / RECORDS ARE } { STANDARD / OMITTED } [*data-name-1* [, *data-name-2*] ...]

[ ; RECORD CONTAINS [*integer-3* TO *integer-4* CHARACTERS] ]

; { REPORT IS / REPORTS ARE } *report-name-1* [, *report-name-2*] ...

[ VALUE OF *data-name-3* IS { *data-name-4* / *literal-1* } ]

[ , *data-name-5* IS { *data-name-6* / *literal-2* } ] ... ]

##### Description:

The File Description furnishes information concerning the physical structure, identification, and record names pertaining to a given file.

The level indicator FD identifies the beginning of a File Description and must precede the file-name.

All semicolons are optional in the File Description but the entry must be terminated by a period.

The clauses which follow the name of the file are optional in many cases, and their order of appearance is immaterial.

## A.2.1.1. Standard COBOL Clauses

The BLOCK, LABEL, RECORD, and VALUE clauses are used as specified in the standard COBOL. The only addition required for using the Report Writer is the REPORT clause.

## A.2.1.2. REPORT Clause

**Format:**

$\left. \begin{array}{l} \text{REPORT IS} \\ \text{REPORTS ARE} \end{array} \right\} \text{report-name-1 [ , report-name-2] . . .}$

**Description:**

A REPORT clause is required in the FD entry to list the names of the reports to be produced.

The REPORT clause cross references the Report Description entries with their associated File Description entry.

Each report-name listed in the FD entry must be the subject of a Report Description (RD) entry in the Report Section.

The presence of more than one report-name indicates that the file contains more than one report. The order in which they are listed is not significant.

## A.2.1.3. The Report Section

A report is described in a section headed by the keywords REPORT SECTION. The Report Section appears after the Working-Storage Section if one is specified.

The Report Section consists of two types of entries for each report: one describes the physical aspects of the report format; the other type describes conceptual characteristics of the items which make up the report and their relation to the report format. These are:

- (a) Report Description (RD)
- (b) Report-group description entries

The Report Description entry contains information pertaining to the overall format of a report named in the File Section and is uniquely identified in the Report Section by the level indicator RD. The characteristics of the report page are provided by describing the number of physical lines per page and the limits for presenting specified headings, footings, and details within a page structure. Data items which act as control factors during presentation of the report are specified in the RD entry. Each report named in an FD entry in the File Section must be defined by an RD entry.

A report may be divided into report groups. A report group is a set of data items that is to be presented as an individual unit, irrespective of its physical format structure. It may consist of several report lines containing many data items or of one report line containing a single data item. Three categories of report group definitions are provided: heading groups, footing groups, and detail groups.

The data items constituting a report group must be identified by the level number 01 and a TYPE clause. Report group names are required when reference is made in the Procedure Division:

- (a) to a detail report group by a GENERATE statement
- (b) to a heading or footing report group by a USE statement.

The description of the report group, analogous to that of the data record, consists of a set of entries defining the characteristics of the elements. The placement of an item in relation to the entire report group and to the overall report format, the format description of all items, and any control factors associated with the group are defined by the entry.

#### A.2.1.4. Report Description Entries

##### Format:

```
RD report-name
[; CODE mnemonic-name-1]
[ { CONTROL IS } { FINAL
  { CONTROLS ARE } { identifier-1 [, identifier-2]. . . } ]
[; PAGE [ LIMIT IS
  [ LIMITS ARE ] integer-1 { LINE
  { LINES } [, HEADING integer-2]
[, FIRST DETAIL integer-3] [, LAST DETAIL integer-4]
[, FOOTING integer-5] ]
```

##### Description:

The Report Description furnishes information concerning the physical layout of each page of a report (PAGE clause) and provides information required for control break operations (CONTROL clause).

The level indicator RD identifies the beginning of a Report Description and must precede the report-name. The report-name must appear in a REPORT clause of an FD entry. The clauses which follow the report-name are optional in many cases, and their order of appearance is immaterial. Semicolons are optional in the Report Description but the entry must be terminated by a period.

#### A.2.1.4.1. CODE Clause

##### Format:

```
[; CODE mnemonic-name-1]
```

##### Description:

The CODE clause provides identification of a report when more than one report is to be transferred to an output device. Mnemonic-name-1 allows selection of one report from a number of reports which may exist on an output medium. The rules for forming the mnemonic-name are described in the manual for a specific implementation.

## A.2.1.4.2. CONTROL Clause

**Format:**

$$\left. \begin{array}{l} \{ \underline{\text{CONTROL IS}} \\ \{ \underline{\text{CONTROLS ARE}} \} \end{array} \right\} \left\{ \begin{array}{l} \underline{\text{FINAL}} \\ \text{identifier-1 [ , identifier-2] ...} \\ \underline{\text{FINAL}} , \text{identifier-1 [ , identifier-2] ...} \end{array} \right\}$$
**Description:**

The CONTROL clause provides information indicating the following:

- (a) the name of the identifier which initiates a control break when a change in the value occurs.
- (b) the order in which control breaks are to occur when more than one control break is to be executed.

The identifiers must be defined in the File or Working-Storage Section of the Data Division.

The CONTROL clause is required when control headings or control footing report groups are specified.

The identifiers specified in the CONTROL clause are the only identifiers referred to by the RESET and TYPE clauses in a Report Group Description entry for this report.

The identifiers specify the control hierarchy for this report and are listed in order from major to minor; FINAL is the highest control, identifier-1 is the major control, identifier-2 is the intermediate control, etc. The last identifier specified is the minor control.

## A.2.1.4.3. PAGE LIMIT Clause

**Format:**

$$\left[ \underline{\text{PAGE}} \left[ \begin{array}{l} \underline{\text{LIMIT IS}} \\ \underline{\text{LIMITS ARE}} \end{array} \right] \text{integer-1} \left\{ \begin{array}{l} \underline{\text{LINE}} \\ \underline{\text{LINES}} \end{array} \right\} [ , \underline{\text{HEADING}} \text{integer-2}] \right. \\ \left. [ , \underline{\text{FIRST DETAIL}} \text{integer-3}] [ , \underline{\text{LAST DETAIL}} \text{integer-4}] \right. \\ \left. [ , \underline{\text{FOOTING}} \text{integer-5}] \right]$$
**Description:**

The PAGE LIMIT clause indicates the specific line control to be maintained within the logical presentation of a page. The clause is required when page format must be controlled by the Report Writer. The PAGE LIMIT clause may be omitted when no association is desired between report groups and the physical format of an output page.

Integers specified in the PAGE LIMIT clause must be positive. Integer-2, integer-3, integer-4, and integer-5 must be either less than or equal to integer-1.



Integer-1 clause is required to specify the depth of the report page; the depth of the report page may or may not be equal to the physical perforated continuous form often associated in a report with the page length.

The value specified in the PAGE clause must not exceed the maximum value allowed for the line counter in a given implementation.


No print group will start preceding integer-2. If HEADING integer-2 is omitted, a value of 1 (line number 1) is assumed.

FIRST DETAIL integer-3 indicates the upper boundary for printing body groups (*body* refers to a control heading, control footing, or a detail report group). No detail or control print group will be printed before integer-3. If FIRST DETAIL integer-3 is omitted, integer-3 is considered to be equivalent to the value of integer-2.

LAST DETAIL integer-4 indicates the lower boundary for printing a CH or detail report group. Detail or control heading print group will not extend beyond integer-4. If LAST DETAIL integer-4 is omitted, integer-4 is considered to be equivalent to the value of integer-5.

FOOTING integer-5 indicates the lower limit of the page for printing footing groups. A control footing print group will not start before integer-3 or extend beyond integer-5. A page footing group will follow integer-5. If FOOTING integer-5 is omitted, integer-5 is considered to be equivalent to the value of integer-4. If both LAST DETAIL integer-4 and FOOTING integer-5 are omitted, integer-4 and integer-5 are both considered to be equivalent to the value of integer-1.

The following chart pictorially represents page format report group control when the PAGE LIMIT clause is specified:



Absolute LINE NUMBER or absolute NEXT GROUP spacing must be consistent with controls specified in the PAGE LIMIT clause.

Only one PAGE LIMIT clause may be specified per Report Description entry.

## A.2.2. Report Group Entries

**Formats:***Format 1:*01 [*data-name-1*]
$$\left[ ; \underline{\text{LINE NUMBER}} \text{ IS } \left\{ \begin{array}{l} \textit{integer-1} \\ \underline{\text{PLUS}} \textit{integer-2} \\ \underline{\text{NEXT PAGE}} \end{array} \right\} \right]$$

$$\left[ ; \underline{\text{NEXT GROUP}} \text{ IS } \left\{ \begin{array}{l} \textit{integer-3} \\ \underline{\text{PLUS}} \textit{integer-4} \\ \underline{\text{NEXT PAGE}} \end{array} \right\} \right]$$

$$\left[ ; \underline{\text{TYPE}} \text{ IS } \left\{ \begin{array}{l} \underline{\text{REPORT HEADING}} \\ \underline{\text{RH}} \\ \underline{\text{PAGE HEADING}} \\ \underline{\text{PH}} \\ \left\{ \underline{\text{CONTROL HEADING}} \right\} \left\{ \textit{integer-1} \right\} \\ \left\{ \underline{\text{CH}} \right\} \left\{ \underline{\text{FINAL}} \right\} \\ \underline{\text{DETAIL}} \\ \underline{\text{DE}} \\ \left\{ \underline{\text{CONTROL FOOTING}} \right\} \left\{ \textit{identifier-2} \right\} \\ \left\{ \underline{\text{CF}} \right\} \left\{ \underline{\text{FINAL}} \right\} \\ \underline{\text{PAGE FOOTING}} \\ \underline{\text{PF}} \\ \underline{\text{REPORT FOOTING}} \\ \underline{\text{RF}} \end{array} \right\} \right]$$

$$\left[ ; \underline{\text{USAGE}} \text{ IS } \underline{\text{DISPLAY}} \right]$$
*Format 2:**level-number* [*data-name-1*][; BLANK WHEN ZERO][; COLUMN NUMBER IS *integer-1*][; GROUP INDICATE]
$$\left[ ; \left\{ \begin{array}{l} \underline{\text{JUSTIFIED}} \\ \underline{\text{JUST}} \end{array} \right\} \text{ RIGHT} \right]$$

$$\left[ ; \underline{\text{LINE NUMBER}} \text{ IS } \left\{ \begin{array}{l} \textit{integer-2} \\ \underline{\text{PLUS}} \textit{integer-3} \\ \underline{\text{NEXT PAGE}} \end{array} \right\} \right]$$

$$\left[ ; \left\{ \begin{array}{l} \underline{\text{PICTURE}} \\ \underline{\text{PIC}} \end{array} \right\} \text{ IS } \textit{character-string} \right]$$

$$\left[ ; \underline{\text{RESET ON}} \left\{ \begin{array}{l} \textit{identifier-1} \\ \underline{\text{FINAL}} \end{array} \right\} \right]$$

$$\left[ ; \underline{\text{SOURCE}} \text{ IS } \textit{identifier-2} \right. \\ \left. ; \underline{\text{SUM}} \textit{identifier-3} \left[ , \textit{identifier-4} \right] \dots \left[ \underline{\text{UPON}} \textit{data-name-2} \right] \right\} \\ \left[ ; \underline{\text{VALUE}} \text{ IS } \textit{literal-1} \right. \\ \left. ; \underline{\text{USAGE}} \text{ IS } \underline{\text{DISPLAY}} \right]$$

**Description:**

The Report Group Description entry specifies the characteristics of a particular report group and of the individual data items within a report group.

All semicolons are optional in the Report Group Description but the entry must be terminated by a period. Except for the data-name clause, which, when present, must immediately follow the level-number, the clauses may be written in any order.

Format 1 is used to indicate a report group; the report group extends from this entry to the next report group level 01 entry. Format 2 is used to indicate an elementary item or group item within a report group. The level-number may be any number from 1-49. If a report group is an elementary entry, Format 2 may include the TYPE and NEXT GROUP clauses in order to specify the report group and elementary item in the same entry.

The use of BLANK WHEN ZERO, PICTURE, and USAGE clauses are the same as in standard COBOL and are not discussed here.

## A.2.2.1. Data-name Clause

**Format:**

[*data-name-1*]

**Description:**

Data-name specifies the name of a group item or an elementary item. It is required in a Report Group Description entry when the name is to be referenced in the Procedure Division of the Report Section. The data-name must be specified at an 01 level when a report group is to be referenced in a GENERATE statement (for a DE group item), or a USE statement (for heading and footing groups). A data-name is required for an item specifying a SUM clause when the sum counter is to be referenced. The data-name for the sum counter may be specified at any level.

## A.2.2.2. LINE NUMBER Clause

**Format:**

$$\left[ ; \underline{\text{LINE NUMBER IS}} \left\{ \begin{array}{l} \textit{integer-1} \\ \underline{\text{PLUS}} \textit{integer-2} \\ \underline{\text{NEXT PAGE}} \end{array} \right\} \right]$$
**Description:**

The LINE NUMBER clause specifies the line position on a page where a line is to be printed. The line number may be an absolute value (*integer-1*), a value relative to the previous line number (*integer-2*), or the LINE NUMBER clause may specify that a page break is to occur (NEXT PAGE).

Integer-1 and integer-2 must be positive integers. Integer-1 must be within the range specified in the PAGE LIMITS clause.

A **LINE NUMBER** clause must be given for each report line of a report group. A line number can be specified at either a group or an elementary item level. A line number at a subordinate level may not contradict a line number specified at a group level.

If the **LINE NUMBER** clause is specified at the report group level, entries for the first report line within the report group are presented on the specified line number. If **LINE NUMBER** is specified for an entry on other than the report group level, sequential entries following that entry within the report group with the same level number are implicitly presented on the same **LINE NUMBER**.

The absolute line number entries must be specified in ascending order. An absolute **LINE NUMBER** cannot be preceded by a relative **LINE NUMBER**.

The **NEXT PAGE** phrase may be used to indicate an automatic skip to the next page before presenting the first line of the current Report Group. Appropriate page footings and page headings will be produced as specified.

#### A.2.2.3. **NEXT GROUP** Clause

**Format:**

$$\left[ \underline{\text{NEXT GROUP}} \text{ IS } \left\{ \begin{array}{l} \text{integer-1} \\ \underline{\text{PLUS}} \text{ integer-2} \\ \underline{\text{NEXT PAGE}} \end{array} \right\} \right]$$

The **NEXT GROUP** clause is used to specify an automatic skip to the next page after the last line of the current report group is printed. Appropriate page footings and page headings will be produced as specified.

Integer-1 and integer-2 must be positive integers. Integer-1 must not exceed the value specified in the **PAGE LIMIT** specification.

Integer-1 indicates an absolute value for a line number. The line counter is set to the value specified after the last line of the report group containing the clause is produced.

Integer-2 indicates a relative line number which increments the line counter by the integer-2 value. It represents the number of lines skipped following the last line of the current report group. (Further spacing is specified by the **LINE NUMBER** clause of the next report group produced.)

The **NEXT GROUP** clause must appear only at the 01 level which defines the report group. When specified for a control footing/heading report group, the **NEXT GROUP** clause results in automatic line spacing only when a control break occurs on the level for which that control is specified.

## A.2.2.4. TYPE IS Clause

**Format:**

;TYPE IS	{	<u>REPORT HEADING</u>	
		<u>RH</u>	
		<u>PAGE HEADING</u>	
		<u>PH</u>	
		{ <u>CONTROL HEADING</u> }	{ <i>identifier-n</i> }
		{ <u>CH</u> }	{ <u>FINAL</u> }
		<u>DETAIL</u>	
		<u>DE</u>	
		{ <u>CONTROL FOOTING</u> }	{ <i>identifier-n</i> }
		{ <u>CF</u> }	{ <u>FINAL</u> }
<u>PAGE FOOTING</u>			
<u>PF</u>			
<u>REPORT FOOTING</u>			
<u>RF</u>			

**Description:**

The TYPE clause identifies a particular type of report group. It must appear at an 01 level and indicates the time for generation of a report group in relation to other specified groups.

If the report group is described as other than detail, its generation is an automatic Report Writer function. If a detail report group is described, a GENERATE data-name statement directs the Report Writer to produce the named report group.

The REPORT HEADING entry indicates a report group that is produced only once at the beginning of a report. No printing precedes the RH report group. There can be only one report group of this type in a report.

The PAGE HEADING entry indicates a report group that is automatically produced at the top of each page of the report. There can be only one report group of this type in a report.

The CONTROL HEADING entry indicates a report group that is produced at the beginning of a control group for a designated identifier; in the case of FINAL, the CH is produced once before the first control group at the initiation of a report during the execution of the first GENERATE statement. There can be only one report group of this type for each identifier and for the FINAL specified in a report. In order to produce control heading report groups other than FINAL, a control break must occur. SOURCE clauses used in the CONTROL HEADING FINAL report group refer to the values at the time the first GENERATE statement is executed.

The DETAIL entry indicates a report group that is produced for each GENERATE statement in the Procedure Division.

The CONTROL FOOTING entry indicates a report group that is produced at the end of a control group for a designated identifier or, in the case of FINAL, is produced once at the termination of a report. There can be only one report group of this type for each identifier and for the FINAL entry specified in a report. In order to produce any CONTROL FOOTING report groups, a control break must occur, or the TERMINATE statement executed.

The PAGE FOOTING entry indicates a report group that is automatically produced at the bottom of each page of the report. There can be only one report group of this type in a report.

The REPORT FOOTING entry indicates a report group that is produced only once at the termination of a report. No printing follows an RF report group. There can be only one report group of this type in a report. SOURCE clauses used in report footing report groups refer to the value of items at the time the TERMINATE statement is executed.

The heading or footing report groups occur in the following Report Writer sequence if all exist for a given report:

REPORT HEADING (one occurrence only)  
PAGE HEADING  
.  
.  
CONTROL HEADING  
DETAIL  
CONTROL FOOTING  
.  
.  
PAGE FOOTING  
REPORT FOOTING (one occurrence only)

Control heading report groups are presented in the following hierarchical arrangement:

Final Control Heading  
Major Control Heading  
.  
.  
Minor Control Heading

Control footing report groups are presented in the following hierarchical arrangement:

Minor Control Footing  
.  
.  
Major Control Footing  
Final Control Footing

Control heading report groups appear with the current values of any indicated SOURCE data items. Control footing report groups appear with the previous values of any indicated control source data items. The preceding rules remain the same when a USE procedure is specified for a control footing report group.

#### A.2.2.5. COLUMN NUMBER Clause

**Format:**

[; COLUMN NUMBER IS *integer-1*]

**Description:**

The COLUMN NUMBER clause indicates the absolute column number of the printed page of the higher-order (leftmost) character of the elementary item. Within a report group and a particular line number specification, column number entries must be indicated from left to right.

The COLUMN NUMBER clause can only be given at the elementary level within a report group. If the column number is not specified, the printing of the item is suppressed when the report group is produced.

#### A.2.2.6. GROUP INDICATE Clause

**Format:**

[; GROUP INDICATE]

**Description:**

The GROUP INDICATE clause indicates that the associated elementary item is to be produced only on the first occurrence of the item after any control break or page break. It is specified only with an elementary item in a detail report group.

#### A.2.2.7. RESET Clause

**Format:**

[; RESET ON  $\left. \begin{array}{l} \textit{identifier-1} \\ \text{FINAL} \end{array} \right\}$ ]

**Description:**

The RESET clause is used to allow progressive totaling of subtotals by inhibiting the automatic reset of a counter when a CF report group is produced.

A sum counter is automatically reset to zero after the generation of the control footing in which the counter is defined. The RESET clause allows for changing the automatic condition by specifying the point at which resetting is to occur.

Identifier-1 must be an identifier specified in the CONTROL clause. It must be a higher level identifier than the one associated with the control footing report group item containing the SUM and RESET clauses.

The RESET clause is used only in conjunction with the SUM clause.

## A.2.2.8. SOURCE, SUM, and VALUE Clauses

**Format:**

```
{ ; SOURCE IS identifier-2
  ; SUM identified-3 [, identifier-4] ... [UPON data-name 2] }
  ; VALUE IS literal-1
```

**Description:**

The SOURCE, SUM, or VALUE clause defines the purpose of an elementary item within a report group.

Identifier-2, identifier-3, and identifier-4 must represent items appearing in the File or Working-Storage Section or sum counters in the Report Section.

Literal-1 must be numeric, nonnumeric, or a figurative constant.

The SOURCE clause indicates a data item that is to be presented on the print line. This data item is called a source data item or source item. The item is presented according to the PICTURE clause in the associated elementary report group entry.

The SUM clause defines a field to be used for summing. The sum is automatically presented during a control break. It may only appear in a control footing report group.

If a SUM counter is referred to by a Procedure Division statement or Report Section entry, a data-name must be specified with the SUM clause entry. The data-name then represents the summation counter.

The counter is automatically generated by the Report Writer to total the operands specified immediately following the SUM.

A SUM counter is algebraically incremented just before presentation of the detail report group in which the item being summed appears as a SOURCE item.

An entry containing a SUM clause must also specify a PICTURE clause. The PICTURE clause may contain editing information. Editing occurs when the value of the sum counter is presented; at other times, the value is treated as a numeric data item. The sum counter must be large enough to accommodate the summed quantity without truncation of integral digits.

Each item being summed, that is, identifier-3, identifier-4, etc., must appear as a source item in a detail report group, or be the name of a sum counter entry in a control footing report group at an equal or lower level in the control hierarchy.

Although the items must be explicitly written in a detail report group, they may be actually suppressed at presentation time. In this manner, direct association without ambiguity can be made from the current data available by a GENERATE statement to the data items to be presented within the Report Section.

If higher level report groups are indicated in the control hierarchy, counter updating, commonly called rolling counters forward, takes place prior to the reset operation.



The summation of data items defined as sum counters in control footing report groups is accomplished explicitly or implicitly with the Report Writer automatically handling the updating function.

The UPON data-name-1 option is required to obtain selective summation for a particular data item which is named as a SOURCE item in two or more detail report groups. Identifier-3 and identifier-4 must be SOURCE data items in data-name-1, where data-name-1 is the name of a detail report group. If the UPON data-name-1 option is not used, identifier-3, identifier-4, etc., respectively, are added to the SUM counter at each execution of a GENERATE statement. This statement generates a detail report group that contains the sum operands at the elementary level.

The VALUE clause describes a numeric, nonnumeric, or literal to be printed on a line.

#### A.2.2.9. LINE-COUNTER and PAGE-COUNTER Fixed Data-Names

**Format:**

PAGE-COUNTER

LINE-COUNTER

**Description:**

PAGE-COUNTER

The data-name PAGE-COUNTER refers to a storage location used for maintaining a page count. The counter is provided by the Report Writer when the word PAGE-COUNTER is specified as a source data-item. The size of the counter is determined by the information specified in the PICTURE clause associated with the SOURCE clause. Sufficient positions must be allocated for the counter to prevent overflow.

One page counter is supplied for each report. If more than one report description entry exists in the Report Section, the user must qualify PAGE-COUNTER by the report name when a reference is required in the Procedure Division.

Initially, the page counter is set to one by the Report Writer. If a starting value other than one is desired, the value in the counter can be changed by procedural commands after an INITIATE statement is executed.

The page counter is automatically incremented by one each time a page break occurs. Incrementing takes place after the production of any page footing report group but before production of any page heading report group.

LINE-COUNTER

The data-name LINE-COUNTER refers to a storage location used by the Report Writer for maintaining a line count. A line counter is provided by the Report Writer when the word LINE-COUNTER is specified as a source data-item. The size of the counter is determined by the information specified in the PICTURE clause.

The line counter data-name may be referenced in the SOURCE clause of the Data Division or in statements in the Procedure Division. Changing the value in the counter by Procedure Division statements may cause the Report Writer to lose control of the page format.

The line-counter name must be qualified when more than one Report Description exists in the Report Section.

The value of the line counter during a Procedure Division test statement represents the number of the last line printed by the previous report group or the number of the last line skipped to as a result of a NEXT GROUP specification.

Writing on a line more than once is prohibited when using Report Writer.

#### A.2.3. Procedure Division

The verbs required in the Procedure Division are INITIATE, GENERATE, and TERMINATE. The declarative USE BEFORE REPORTING is optional.

##### A.2.3.1. GENERATE Statement

**Format:**

GENERATE *identifier*

**Description:**

Identifier must represent a detail report group or an RD entry.

If identifier is the name of a detail report group, the GENERATE statement does all the automatic operations within a Report Writer and produces all necessary report groups on the output medium. This is called detail reporting.

If identifier is the name of an RD entry, the GENERATE statement does all the automatic operations of the Report Writer except it does not print any detail report groups. In this case, all sum counters associated with the report description are algebraically incremented. This is called summary reporting. If more than one detail report group is specified, all sum counters are algebraically incremented each time a GENERATE statement is executed.

A GENERATE statement, implicitly in both detail and summary reporting, produces the following automatic operations (if defined):

- (a) Steps and tests the line counter and/or page counter to produce appropriate page footing and/or page heading report groups.
- (b) Recognizes any specified control breaks to produce appropriate control footing and/or control heading report groups.
- (c) Accumulates into the sum counters all specified identifier(s). Resets the the sum counters on an associated control break.
- (d) Executes any specified routines defined by a USE statement before generation of the associated report group(s).

During the execution of the first GENERATE statement the following report groups associated with the report, if specified, are produced in the order specified:

- (a) report heading report group
- (b) page heading report group
- (c) all control heading report groups in the order FINAL, major or minor
- (d) the detail report group, if specified in the GENERATE statement

If a control break is recognized at the time of execution of a GENERATE statement (other than the first that is executed for a report), all control footing report groups specified for the report are produced from the minor report group up to and including the report group specified for the identifier which caused the control break. Then, the control heading report groups specified for the report, from the report group specified for the identifier that caused the control break down to the minor report groups, are produced in that order. The detail report group specified in the GENERATE statement is then produced.

#### A.2.3.2. INITIATE Statement

**Format:**

INITIATE *report-name-1* [, *report-name-2*] ...

**Description:**

The INITIATE statement begins processing of a report. Each report-name must be defined by a Report Description entry in the Report Section of the Data Division.

The INITIATE statement causes the Report Writer to reset all the sum counters to zero, set the page counter location to 1, and the line counter location to 0.

The INITIATE statement does not open the file with which the report is associated. An OPEN statement for the file must be executed before the INITIATE statement.

A second INITIATE statement for a particular report-name may not be executed unless a TERMINATE statement has been executed for that report-name subsequent to the first INITIATE statement.

#### A.2.3.3. TERMINATE Statement

**Format:**

TERMINATE *report-name-1* [, *report-name-2*] ...

**Description:**

The TERMINATE statement terminates the processing of a report.

Each report-name given in a TERMINATE statement must be defined by an RD entry in the Data Division.

The TERMINATE statement produces all the control footings associated with the last GENERATE statement as if a control break had just occurred at the highest level, and completes the Report Writer function for the named reports. The TERMINATE statement also produces the last page footing and the report footing report groups associated with the report.

A second TERMINATE statement for a particular report may not be executed unless a second INITIATE statement has been executed for the report-name. If a TERMINATE statement has been executed for a report, a GENERATE statement for that report must not be executed unless an intervening INITIATE statement for that report is executed.

SOURCE clauses used in control footing FINAL or report footing report groups refer to the values of the items during execution of the TERMINATE statement.

#### A.2.3.4. USE BEFORE REPORTING Clause

**Format:**

USE BEFORE REPORTING *identifier-1*

**Description:**

The USE BEFORE REPORTING statement specifies Procedure Division statements that are executed just before a report group named in the Report Section of the Data Division is produced. The USE statement itself is never executed; rather it defines the conditions calling for the execution of the USE procedures.

Identifier-1 represents a report group named in the Report Section of the Data Division. An identifier must not appear in more than one USE statement.

A USE statement, when present, must immediately follow a section header in the Declarative portion of the Procedure Division and must be followed by a period followed by a space. The remainder of the section must consist of one or more procedural paragraphs that define the procedures to be used.

The GENERATE, INITIATE, or TERMINATE statements must not appear in the Declarative Section.

The designated procedures are executed by the Report Writer just before the named report is produced, regardless of page break or control break associations with report groups. The report group may be any type except detail.

Within a USE procedure, there must not be any reference to any nondeclarative procedures. Conversely, in the nondeclarative portion there must be no reference to procedure-names that appear in the declarative portion, except that PERFORM statements may refer to a USE BEFORE REPORTING declarative or to the procedures associated with the USE declarative.



